



DESENVOLVIMENTO DE SOFTWARE MULTIPLATAFORMA

Disciplina: IAL-010 Algoritmos e Lógica de Programação

Aula 06: ESTRUTURAS CONDICIONAIS

Data 02/04/2024

Prof. Me. Anderson Silva Vanin

Estruturas Condicionais

As estruturas de seleção ou decisão são utilizadas quando existe a necessidade de verificar condições para a realização de uma instrução ou de uma sequência de instruções. Os testes de seleção também podem ser utilizados para verificar opções de escolha.

Estruturas Condicionais

Exemplo:

Suponha que uma pessoa esteja jogando um jogo de computador:


1. Para que o jogador passe de uma fase (etapa) para a fase seguinte, é necessário que se verifique se ele atingiu a pontuação exigida. Assim, existe uma condição para a realização de uma sequência de instruções para liberar o acesso à próxima fase do jogo.
2. Ao final do jogo, uma pergunta é feita: “Deseja continuar jogando?”. O jogador poderá escolher entre as respostas sim ou não. Uma decisão é o resultado da avaliação de uma expressão booleana. Toda condição pode ser encarada como uma pergunta que pode ter a resposta verdadeiro (.v.) ou falso (.f.). As estruturas de seleção podem ser do tipo simples, composto ou encadeado.

Pseudocódigo

A sintaxe básica da decisão unidirecional é:

```
se (condição) então [início do bloco de decisão]  
    conjunto de instruções  
fim-se [fim do bloco de decisão]
```


O conjunto de instruções só será executado se a condição for verdadeira. Uma condição é uma comparação que possui dois valores possíveis, verdadeiro ou falso.



Exemplos

1. Verificar se um número fornecido pelo usuário é ímpar. Se for, exibir a mensagem “O número informado é ímpar”.

```
algoritmo "VerificaNumeroImpar"  
  // Função : Verificar se um número informado é ímpar  
  // Autor : Anderson Vanin  
  // Data : 01/04/2024  
  // Seção de Declarações  
  var  
    numero: inteiro  
  inicio  
    escreval("Digite um número inteiro:")  
    leia(numero)  
    se (numero mod 2 <> 0) entao  
      escreval("O número informado é ímpar!")  
    fimse  
  finalgoritmo
```




```
Digite um número inteiro:  
5  
O número informado é ímpar!  
  
*** Fim da execução.  
*** Feche esta janela para retornar ao Visualg.
```

Exemplos

2. Determine o maior de dois números dados. Suponha que o primeiro deles é o maior, armazenando-o em uma variável MAIOR e depois, compare se o maior procurado é o segundo dos números dados, neste caso o conteúdo da variável MAIOR deve ser alterado.

```
algoritmo "VerificaMaiorNumero"  
  // Função : Verificar o maior número de 2  
  // Autor : Anderson Vanin  
  // Data : 01/04/2024  
  // Seção de Declarações  
  var  
  a,b,maior:inteiro  
  inicio  
    escreval("Digite um número inteiro:")  
    leia(a)  
    escreval("Digite outro número inteiro:")  
    leia(b)  
    maior <- a  
    se (b>a) entao  
      maior <- b  
    fimse  
    escreval("O maior dos números",a," e ",b," é ",maior)  
  finalgoritmo
```



```
Digite um número inteiro:  
5  
Digite outro número inteiro:  
9  
O maior dos números 5 e 9 é 9  
  
*** Fim da execução.  
*** Feche esta janela para retornar ao Visualg.
```

Estrutura Condicional Composta

A estrutura de seleção composta prevê dois conjuntos de instruções para serem realizados de acordo com a avaliação da condição: um conjunto de instruções que será realizado quando a condição resultar verdadeiro e um conjunto de instruções para resultado falso.

se (condição) **então** [início do bloco de decisão]

conjunto de instruções a

[conjunto de instruções que será realizado se o teste de condição resultar verdadeiro]

senao


conjunto de instruções b

[conjunto de instruções que será realizado se o teste de condição resultar falso]

fim-se [fim do bloco de decisão]

Estrutura Condicional Composta

Fluxograma



Exemplos


1. A empresa *XSoftwares* concedeu um bônus de 20% do valor do salário a todos os funcionários com tempo de trabalho na empresa igual ou superior a cinco anos, e 10% aos demais. Calcular e exibir o valor do bônus.

Para resolver o problema é necessário conhecer o valor do salário e o tempo de serviço do funcionário.

Exemplos

```
algoritmo "BonusSalario"

var
salario,bonus:real
tempo:inteiro
inicio
escreval("Digite o salário do funcionário:")
leia(salario)
escreval("Digite o tempo de trabalho:")
leia(tempo)
se (tempo>=5) entao
    bonus <- salario * 0.20
senao
    bonus <- salario * 0.10
fimse
escreval("O valor do bônus é de R$ ",bonus)
finalgoritmo
```



```
Digite o salário do funcionário:
1200
Digite o tempo de trabalho:
6
O valor do bônus é de R$ 240

*** Fim da execução.
*** Feche esta janela para retornar ao Visualg.
```

Estrutura Condicional Encadeada

Uma estrutura de seleção encadeada é uma sequência de testes de seleção, os quais serão executados ou não de acordo com o resultado das condições e com o encadeamento dos testes.

Um teste de seleção pode ter dois conjuntos de instruções, um para resultado verdadeiro e outro para falso, porém esses conjuntos de instruções podem conter outros testes de seleção, que, por sua vez, também podem conter outros, e assim por diante.


Estrutura Condicional Encadeada

```
se (condição_1) então
    se (condição_2) então
        conjunto de instruções a
    senao
        conjunto de instruções b
    fim-se
senao
    conjunto de instruções c
fim-se
```

Estrutura Condicional Encadeada

Se a condição_1 resultar *verdadeiro*, então será realizado o teste da condição_2; se esse teste resultar *verdadeiro*, será realizado o conjunto de instruções A; se resultar *falso*, será realizado o conjunto de instruções B. Se o teste da condição_1 resultar *falso*, será realizado o conjunto de instruções C.

Fluxograma




Exemplos

1. Sabemos que um aluno é aprovado caso apresente média maior ou igual a 7.0 e frequência maior ou igual a 75%. Na verdade, em uma situação real, se o aluno obtiver a frequência mínima exigida e uma média entre 3 e 7, ainda teria direito a uma última avaliação de recuperação.

Exemplos

```
algoritmo "VerificaAprovacaoAluno"  
var  
frequencia,media:real  
inicio  
escreval("Digite a média do aluno:")  
leia(media)  
escreval("Digite a frequência do aluno:")  
leia(frequencia)  
se (frequencia>=75) entao  
    se media >=7 entao  
        escreval("APROVADO")  
    senao  
        se media>=3 entao  
            escreval("RECUPERAÇÃO")  
        senao  
            escreval("REPROVADO POR MEDIA")  
        fimse  
    fimse  
senao  
    escreval("REPROVADO POR FALTAS")  
fimse  
finalgoritmo
```



```
Digite a média do aluno:  
6  
Digite a frequência do aluno:  
70  
REPROVADO POR FALTAS  
  
*** Fim da execução.  
*** Feche esta janela para retornar ao Visualg.
```

Estruturas Condicionais de Múltipla Escolha

Uma estrutura de seleção de múltipla (ou estrutura de seleção homogênea) escolha funciona como um conjunto de opções para escolha. Existem duas maneiras para representá-la:

- Utilizando o encadeamento da instrução Se;
- Utilizando a instrução escolha caso.

A segunda opção é a mais indicada.

Estruturas Condicionais de Múltipla Escolha

Estrutura com condicionais encadeadas:


```
se (variável = tal_coisa_1) então  
    faça conjunto de instruções A  
senão  
    se (variável = tal_coisa_2) então  
        faça conjunto de instruções B  
    senão  
        se (variável = tal_coisa3) então  
            faça conjunto de instruções C  
        senão  
            faça conjunto de instruções D  
    fim-se  
fim-se  
fim-se
```

Estrutura com seleção de múltipla escolha

```
escolha variável  
    caso tal_coisa_1:  
        faça conjunto de instruções A  
    caso tal_coisa_2  
        faça conjunto de instruções B  
    caso tal_coisa_3  
        faça conjunto de instruções C  
    caso contrário  
        faça conjunto de instruções D  
fim-escolha
```

Estruturas Condicionais de Múltipla Escolha

Fluxograma




Exemplos

1. Leia o código de um produto e exiba seu nome de acordo com a tabela a seguir.

<i>Código do produto</i>	<i>Nome do produto</i>
001	Caderno
002	Lápis
003	Borracha
Qualquer outro	Diversos

Exemplos

```
algoritmo "Produto"  
var  
codigo:inteiro  
inicio  
escreval("Digite o código do produto:")  
leia(codigo)  
escolha codigo  
    caso 1  
        escreval("O produto é CADERNO")  
    caso 2  
        escreval("O produto é LÁPIS")  
    caso 3  
        escreval("O produto é BORRACHA")  
    outrocaso  
        escreval("DIVERSOS")  
fimescolha  
fimalgoritmo
```



```
Digite o código do produto:  
3  
O produto é BORRACHA  
  
*** Fim da execução.  
*** Feche esta janela para retornar ao Visualg.
```

Exercícios

1. Verifique se um número fornecido pelo usuário é par ou ímpar. Apresente uma mensagem mostrando o número digitado e o resultado do teste.
2. Um aluno realizou três provas de uma disciplina. Considerando o critério abaixo faça um programa que mostre se ele ficou para exame. Em caso positivo, leia a nota do exame e verifique se conseguiu a aprovação ou não.

$$\text{Media} = (\text{prova1} + \text{prova2} + \text{prova3}) / 3$$

A média deve ser maior ou igual a 7,0.

Se não conseguir, a nova média deve ser:

$$\text{Final} = (\text{Media} + \text{Exame}) / 2$$

Nesse caso, a média final deve ser maior ou igual a 5,0.

Exercícios

3. A nota final de um estudante é calculada a partir de três notas atribuídas respectivamente a um trabalho de laboratório, a uma avaliação semestral e a um exame final. A média das três notas mencionadas anteriormente obedece aos pesos a seguir:

<i>Nota</i>	<i>Peso</i>
Trabalho de laboratório	2
Avaliação semestral	3
Exame final	5

Faça um programa que receba as três notas, calcule e mostre a média ponderada e o conceito que segue a tabela abaixo:

<i>Média ponderada</i>	<i>Conceito</i>
8,0 a 10,0	A
7,0 a 8,0	B
6,0 a 7,0	C
5,0 a 6,0	D
0,0 a 5,0	E

Exercícios

4. Faça um programa que recebe três notas de um aluno, calcule e mostre a média aritmética e a mensagem que segue a tabela abaixo. Para alunos de exame, calcule e mostre a nota que deverá ser tirada no exame para aprovação, considerando que a média no exame é 6,0.

<i>Média aritmética</i>	<i>Mensagem</i>
0,0 a 3,0	Reprovado
3,0 a 7,0	Exame
7,0 a 10,0	Aprovado

Exercícios

5. Uma livraria está fazendo uma promoção para pagamento à vista em que o comprador pode escolher entre dois critérios de desconto:

- **Critério A:** R\$ 0,25 por livro + R\$ 7,50 fixo.
- **Critério B:** R\$ 0,50 por livro + R\$ 2,50 fixo.

Faça um programa em que o usuário digite a quantidade de livros que deseja comprar e o programa diga qual é a melhor opção de desconto.

Exercícios

6. O posto de atendimento médico e hospitalar de uma pequena cidade atende em média 138 pessoas por dia e vem observando, ao longo do tempo, que os casos de pessoas com problemas de sobrepeso vêm aumentando a cada ano. Assim, o posto de atendimento determinou que em todos os atendimentos o médico ou o enfermeiro deverão calcular o peso ideal de todos os pacientes atendidos. Elabore um programa que verifique se o paciente está acima de seu peso ideal de acordo com as condições abaixo:
- Para homens: $(72,7 \text{ altura}) - 58$;
 - Para mulheres: $(62,1 \text{ altura}) - 44,7$.